
Why predict Phase 3 readouts?

A model in drug development too long; you never learn whether the model was right

model

dose change new site protocol v3 competitor reorg

outcome?

years later, if ever

Phase 3 readouts

public, dated, binary, dozens per year

predict freeze readout score

The workflow, and where the LLM sits

literature
search

evidence
table (JSON)

Stan model
(cmdstanpy)

PoS,
quantile knots

registry
(frozen)

readout,
score

DOI / PMID / number
verification

daily readout
monitor

human: taxonomy,
priors, weights

LLM session LLM routine, dailyfrozen with the data

LLM session

code deterministic artefact curated, hashed LLM-assisted human-reviewed informs

Freeze before readout: the registry

fit freeze()

prediction_id, evidence_cutoff_date
pos: overall, per-endpoint assurance, rule
endpoints: mean, sd, q025 q25 q50 q75 q975
provenance: git_head, sha256(stan, data)
prospective

readout

resolve()PIT, Brier, log score

append-only; duplicate id refused

git push = timestamp

Hypertrophic Cardiomyopathy (HCM)

ACACIA-HCM: the trial

nHCM
non-obstructive

hypertrophic
cardiomyopathy

aficamten vs placebo
cardiac myosin

inhibitor
1:1, n = 504,

36 weeks

dual co-primary
KCCQ-CSS (points)

peak VO2
(mL/kg/min)

success
either endpoint

𝑝 < 0.05

prior evidence is in obstructive HCM within-arm SD 22 / 5;
MID 5 / 1.0

no multiplicity
adjustment

The transfer question: the same mechanism worked three times in obstructive HCM. Does it
carry to the non-obstructive population, and by how much less?

ACACIA-HCM: frozen April 22, read out May 5

Prior evidence: four oHCM trials

Two prior ingredients

Model part 1: pooling the oHCM trials

𝑦𝑗 ∣ 𝜃𝑗 ∼ 𝒩(𝜃𝑗, 𝜎̃2
𝑗 ) , 𝜎̃𝑗 =

𝜎𝑗
√𝑤𝑗

𝜃𝑗 ∣ 𝜇, 𝜏 ∼ 𝒩(𝜇, 𝜏2)

𝜇 ∼ 𝒩(0, 52), 𝜏 ∼ HalfNormal(1.5)

The discount 𝑤𝑗 widens a trial’s
likelihood; it never moves its estimate.

µ0, σµ

τ0

µ

τ

ηj

θj

yj σ̃j

j = 1, . . . , J

Model part 2: the nHCM predictive

𝜋 ∼ Beta(3.5, 1.5), 𝑍 ∼ Bernoulli(𝜋)

𝛽 ∼ HalfNormal(2), ̃𝜃oHCM ∼ 𝒩(𝜇, 𝜏2)

̃𝜃nHCM = {
̃𝜃oHCM − 𝛽 𝑍 = 1

𝒩(0, 22) 𝑍 = 0

𝜋, 𝑍, 𝛽 have no likelihood term: the transfer is a
prior.

µ τ aπ, bπ

from fitting model

θoHCM β πactive

βsd

Z

θnHCM σnull

if Z = 1:
θoHCM − β

if Z = 0:
Normal(0, σnull)

Success rule: dual co-primary, either hits

single endpoint: Pr(sig ∣ ̃𝜃) = Φ(
̃𝜃

se
− 𝑧0.975) , se = 𝜎𝑤√ 1

252 + 1
252

either: (
̂𝛿KCCQ
̂𝛿VO2

) ∼ 𝒩((
̃𝜃KCCQ
̃𝜃VO2

) , ( se2
1 𝜌 se1se2

𝜌 se1se2 se2
2

)) , 𝜌 = 0.37

PoS = 𝔼[ 𝟙{𝑧KCCQ > 1.96 or 𝑧VO2
> 1.96} ]

Bayesian predictive power: the significance probability averaged over the posterior of the true
effect. 𝜌 is the correlation of the two endpoints’ sampling errors.

Stan: data block

data {
int<lower=1> J; // number of historical oHCM trials (4)
vector[J] y; // placebo-adjusted KCCQ effects
vector<lower=0>[J] sigma; // SEs after discounting: se / sqrt(w)

real mu0; // prior mean of mu (0)
real<lower=0> sigma_mu; // prior SD of mu (5)
real<lower=0> tau0; // HalfNormal scale for tau (1.5)

real<lower=0> beta_sd; // attenuation scale, oHCM -> nHCM (2)
real<lower=0> base_rate_alpha; // Beta prior on Pr(transfer) (3.5)
real<lower=0> base_rate_beta; // (1.5)
real<lower=0> sigma_null; // SD of the null component (2)

}

Stan: the hierarchical part

parameters {
real mu; // oHCM class mean
real<lower=1e-6> tau; // between-trial SD
vector<offset=mu, multiplier=tau>[J] theta; // non-centered

}
model {

mu ~ normal(mu0, sigma_mu); // N(0, 5)
tau ~ normal(0, tau0); // HalfNormal(1.5)
theta ~ normal(mu, tau);
y ~ normal(theta, sigma); // sigma = se / sqrt(w)

}

Stan: the transfer mixture

generated quantities {
real theta_oHCM = normal_rng(mu, tau); // new oHCM trial
real pi_active = beta_rng(base_rate_alpha, base_rate_beta); // prior only
real<lower=0> beta_atten = abs(normal_rng(0, beta_sd)); // prior only
int<lower=0, upper=1> Z = bernoulli_rng(pi_active); // prior only
real theta_nHCM = (Z == 1) ? theta_oHCM - beta_atten

: normal_rng(0, sigma_null);
real lprior = normal_lpdf(mu | mu0, sigma_mu) + normal_lpdf(tau | 0, tau0);
// ... plus y_rep (PPC) and log_lik (LOO)

}

Python: fit, gate, freeze

fit, verdict = fit_stan(STAN_FILE, stan_data, seed=12345, adapt_delta=0.98)
# 0 divergences, R-hat <= 1.01, bulk/tail ESS >= 400 -- or it raises

theta = fit.stan_variable("theta_nHCM")
kccq = build_endpoint_posterior(

"kccq_css", "KCCQ CSS points", draws=theta, mid_threshold=5.0,
beneficial_direction="greater")

freeze(prediction_id="pred:acacia-hcm:20260422:e8a10488", trial=ACACIA_SPEC,
endpoints={"kccq_css": kccq, "peak_vo2": pvo2},
probability_of_success={"overall": 0.572, "success_rule": "either"},
evidence_cutoff_date="2026-04-22", model_version="mixture v0.1", ...)

# append-only; stamps git SHA + SHA-256 of the Stan and data files

How much did we learn?

𝜇 moves from 0 to 6.9 (4.0–9.7); 𝜏 shrinks; 𝜋 does not move at all.

Bayesian Biostatistics 2026 · Leiden 17 / 23 October 2026



Diagnostics and a three-fold backtest

Sampler checks

check value gate

divergences 0 0
max 𝑅̂ 1.003 � 1.01
min bulk ESS 2923 � 400
min tail ESS 2888 � 400

Leave-one-trial-out (oHCM, same model)

held out PoS observed pred → obs

EXPLORER 0.76 success 7.0 → 7.6
VALOR 0.39 success 9.4 → 7.0
SEQUOIA 0.65 success 7.0 → 7.9

4 chains × 2000 draws; PPC: all 4 trials inside the 95% replicate interval. Backtest: Brier 0.18, log
score 0.54, 2/3 inside the 95% predictive — a pipeline check, not calibration.

Sensitivity

All four borrowing weights scaled

0.75× 1.00× 1.25×

𝜇 posterior mean 6.70 6.85 6.95
KCCQ predictive mean 3.60 3.67 3.78
Pr(KCCQ > MID) 0.37 0.40 0.42
PoS, either hits 0.57 0.57 0.58

Readout: scored

KCCQ VO2

predicted mean 3.67 0.83
observed 3.0 0.67
quantile rank 0.40 0.40
𝑧 −0.21 −0.22

Right call (PoS 0.57, success: Brier 0.18, log score 0.56); effects ~22% optimistic; predictive 2–3×
wider than the realized CI.

Market-implied probabilities: a different question

𝑓ℚ(𝑆𝑇) = 𝑒𝑟𝑇 𝜕2𝐶(𝐾, 𝑇 )
𝜕𝐾2 ∣

𝐾=𝑆𝑇

, Pr
ℚ

(𝑆𝑇 > 𝑆∗) = ∫
∞

𝑆∗

𝑓ℚ(𝑆𝑇) 𝑑𝑆𝑇, 𝑆∗ = 1.10×𝑆0

CYTK, options snapshot 2026-04-21 (spot $65.80) Pr(close > $72.38) 80% band

June 18 expiry 0.12 0.04–0.19
July 17 expiry 0.32 0.17–0.47
clinical PoS, either co-primary hits 0.57
realized 2026-05-05 close $77.76, +18%

Risk-neutral, not clinical. The option price carries risk premia, commercial value and timing; it
answers “will the stock clear a threshold by expiry”, not “will the trial hit”.

Bayesian Biostatistics 2026 · Leiden 21 / 23 October 2026



What’s next

registry grows
(queue: 22 Phase 3
readouts in 2026–27)

rank histogram of
quantile ranks
at 𝑛 ≥ 10

(quantile knots stored)

calibration by cohort:
endpoint type, area,
prior-only transfer

freeze every imminent
readout: daily monitor,

self-hosted

one joint
model per trial:

meta-regression +
source study + PoS

long-format evidence
contract shared

across trials

Thank you

Eric Novik
Generable · NYU

eric.novik@nyu.edu

linkedin.com/in/enovik

